
Folds and Fold-Fusion
1. Express the following functions by foldr :

1. all p :: List a → Bool, where p :: a → Bool.

2. elem z :: List a → Bool, where z :: a .

3. concat :: List (List a) → List a .

4. filter p :: List a → List a , where p :: a → Bool.

5. takeWhile p :: List a → List a , where p :: a → Bool.

6. id :: List a → List a .

In case you haven’t seen them, all p xs is True iff. all elements in xs satisfy p, and elem z xs
is True iff. x is a member of xs .

Solution:

1. all p = foldr (𝜆x b → p x ∧ b) True .

2. elem x = foldr (𝜆y b → x = = y ∨ b) False ,

3. concat = foldr (++) [ ] .

4. filter p = foldr (𝜆x xs → if p x then x : xs else xs) [ ] ,

5. takeWhile p = foldr (𝜆x xs → if p x then x : xs else [ ]) [ ] ,

6. id = foldr (:) [ ] .

2. Given p :: a → Bool, can dropWhile p :: List a → List a be written as a foldr?

Solution: No. Consider dropWhile even [5, 4, 2, 1], which ought to be [5, 4, 1, 1].
Meanwhile, dropWhile even [4, 2, 1] = [1], and the lost elements cannot be recov-
ered.

3. Express the following functions by foldr :

1. inits :: List a → List (List a).
2. tails :: List a → List (List a).
3. perms :: List a → List (List a).
4. sublists :: List a → List (List a).
5. splits :: List a → List (List a, List a).

Solution:

1. inits = foldr (𝜆x xss → [ ] :map (x :) xss) [ [ ] ] .

2. tails = foldr (𝜆x xss → (x : head xss) : xss) [ [ ] ] ,

3. perms = foldr (𝜆x xss → concat (map (fan x ) xss)) [ [ ] ]

4. sublists = foldr (𝜆x xss → xss ++map (x :) xss) [ [ ] ]

5. splits can be defined by:

splits = foldr spl [ ( [ ], [ ]) ] ,

where spl x ((xs , ys) : zss) =
( [ ], x : xs ++ ys) :map ((x :) × id ) ((xs , ys) : zss) .

where (f × g) (x , y) = (f x , g y).

4. Prove the foldr -fusion theorem. To recite the theorem: given f ::a → b → b, e ::b, h ::b → c
and g :: a → c → c, we have

h · foldr f e = foldr g (h e) ,

if h (f x y) = g x (h y) for all x and y .

Solution: The aim is to prove that h (foldr f e xs) = foldr g (h e) xs for all xs ,
assuming that h (f x y) = g x (h y).
Case xs := [ ]:

h (foldr f e [ ])
= h e
= foldr g (h e) [ ] .

Case xs := x : xs :

h (foldr f e (x : xs))
= { definition of foldr }
h (f x (foldr f e xs))

= { fusion condition: h (f x y) = g x (h y) }
g x (h (foldr f e xs))

= { induction }
g x (foldr g (h e) xs)

= { definition of foldr }
foldr g (h e) (x : xs) .

5. Prove the map-fusion rule map f ·map g = map (f · g) by foldr -fusion.

Solution: Since map g is a foldr , we proceed as follows:

map f ·map g
= { map g is a foldr }
map f · foldr (𝜆x ys → g x : ys) [ ]

= { foldr -fusion }
foldr (𝜆x ys → f (g x ) : ys) [ ]

= { definition of map as a foldr }
map (f · g) .

The fusion condition is proved below:

map f (g x : ys)
= { definition map }
f (g x ) :map f ys .

6. Prove that sum · concat = sum · map sum by foldr -fusion, twice. Compare the proof with
you previous proof in earlier parts of this course.

Solution:

sum · concat
= sum · foldr (++) [ ]
= { foldr -fusion }
foldr (𝜆xs n → sum xs + n) 0

= { foldr -map fusion, see Exercise 7 }
foldr (+) 0 ·map sum

= sum ·map sum .

Fusion conditions for the foldr -fusion is

sum (xs ++ ys) = sum xs + sum ys ,

which is the key property we needed in the early part of this term to prove the same
property. We have proved the property before, by induction on xs . We omit the proof
here. (Note that we can also prove it by two more foldr -fusion, noting that (++ys) is a
foldr , and so is sum .)

See Exercise 7 for foldr -map fusion. The penultimate equality holds because (+) · sum =

(𝜆xs n → sum xs + n). Instead of foldr -map fusion we cal also use foldr fusion alone.
The fusion condition is sum (sum xs : xss) = sum xs + sum xss .

The foldr -fusion theorem captures the common pattern in these proofs. We only need
to fill in the problem-dependent proofs.

7. Themap fusion theorem is an instance of the foldr -map fusion theorem: foldr f e ·map g =

foldr (f · g) e .
(a) Prove the theorem.

Solution: Since map g is a foldr , we proceed as follows:

foldr f e ·map g
= { map g is a foldr }
foldr f e · foldr (𝜆x ys → g x : ys) [ ]

= { foldr -fusion }
foldr (f · g) (foldr f e [ ])

= { definition of foldr }
foldr (f · g) e .

The fusion condition is proved below:

foldr f e (g x : ys)
= { definition foldr }
f (g x ) (foldr f e ys) .

(b) Express sum ·map (2×) as a foldr .

Solution:

sum ·map (2×)
= foldr (+) 0 ·map (2×)
= { foldr -map fusion }
foldr ((+) · (2×)) 0 .

(c) Show that (2×) · sum reduces to the same foldr as the one above.

Solution:

(2×) · sum
= (2×) · foldr (+) 0
= { foldr fusion }
foldr ((+) · (2×)) 0 .

The fusion condition is

2 × (x + y)
= { distributivity }
2 × x + 2 × y

= { definition of (·) }
((+) · (2×)) x (2 × y) .

8. Prove thatmap f (xs++ys) = map f xs++map f ys by foldr -fusion. Hint: this is equivalent
tomap f · (++ys) = (++map f ys) ·map f . You may need to do (any kinds of) fusion twice.

Solution: Recall that (++ys) is a foldr . Use foldr fusion and foldr -map fusion:

(++map f ys) ·map f
= { foldr -map fusion }
foldr ((:) · f ) (map f ys)

= { foldr fusion }
map f · (++ys) .

The fusion condition of the last step is:

map f (x : zs)
= { definition of map }

f x :map f zs
= { definition of (·) }
((:) · f ) x (map f zs) .

9. Prove that length · concat = sum ·map length by fusion.

Solution: We calculate

length · concat
= length · foldr (++) [ ]
= { foldr -fusion }
foldr ((+) · length) 0

= { sum = foldr (+) 0, foldr -map fusion }
sum ·map length .

The fusion condition is proved below:

length (xs ++ ys)
= { (++) and (+) homorphic }
length xs + length ys

= { definition of (·) }
((+) · length) xs (length ys) .

10. Let scanr f e = map (foldr f e) · tails . Construct, by foldr -fusion, an implementation of
scanr whose number of calls to f is proportional to the length of the input list.

Solution: Recall that tails is a foldr :

tails = foldr (𝜆x xss → (x : head xss) : xss) [ [ ] ] ,

We try to fuse map (foldr f e) into tails . For the base value, notice that

map (foldr f e) [ [ ] ] = [e ] .

To construct the step function, we work on the fusion condition:

map (foldr f e) ((x : head xss) : xss)
= { definition of map }
foldr f e (x : head xss) :map (foldr f e) xss

= { definition of foldr }
f x (foldr f e (head xss)) :map (foldr f e) xss

= { foldr f e (head xss) = head (map (foldr f e) xss) }
let ys = map (foldr f e) xss
in f x (head ys) : ys .

We have therefore constructed:

scanr f e = foldr (𝜆x ys → f x (head ys) : ys) [e ] .

You may find the inductive definition easier to comprehend:

scanr f e [ ] = [e ]
scanr f e (x : xs) = f x (head ys) : ys ,

where ys = scanr f e xs .

11. Recall the function binary :: Nat → [Nat] that returns the reversed binary representation
of a natural number, for example binary 4 = [0, 0, 1]. Also, we talked about a function
decimal :: [Nat] → Nat that converts the representation back to a natural number.

(a) This time, express decimal using a foldr .

Solution:

decimal = foldr (𝜆d n → d + 2 × n) 0 .

(b) Recall the function exp m n =𝑚𝑛. Use foldr -fusion to construct step and base such that

exp m · decimal = foldr step base .

If the fusion succeeds, we have derived a hylomorphism computing 𝑚𝑛:

fastexp m = foldr step base · binary .

Solution: For the base value, we have base = exp m 0 = 1.
For the step function, we calculate

exp m (d + 2 × n)
= { since 𝑚𝑥+𝑦 = 𝑚𝑥 × 𝑚𝑦 }
exp m d × exp m (2 × n)

= { since 𝑚2𝑛 = (𝑚𝑛)2, let square x = x × x }
exp m d × square (exp m n)

= { d is either 0 or 1. Expand the definition }
if d = = 0 then square (exp m n) else m × square (exp m n) .

Therefore we conclude

exp m · decimal = foldr (𝜆d x → if d = = 0 then square x
else m × square x ) 1 .

12. Express reverse :: List a → List a by a foldr . Let revcat = (++) · reverse . Express revcat as a
foldr .

Solution: reverse = foldr (𝜆x xs → xs ++ [x ]) [ ].
To fuse (++) into reverse , the base value is (++) [ ] = id . To construct the step function,
we try to meet the fusion condition:

(++) ((𝜆x xs → xs ++ [x ]) x xs) = step x ((++) xs) .

If we calculate:

(++) ((𝜆x xs → xs ++ [x ]) x xs)
= (++) (xs ++ [x ]) ,

it is hard to figure out how to proceed, since (++) expects another argument. It is easier
to calculate if we supply it another argument ys . We restart and calculate:

(++) ((𝜆x xs → xs ++ [x ]) x xs) ys
= (++) (xs ++ [x ]) ys
= (xs ++ [x ]) ++ ys
= { (++) associative }
xs ++ ([x ] ++ ys)

= { definition of (·) }
(((++) xs) · (x :)) ys

= { factor out x , ((++) xs), and ys }
(𝜆x f → f · (x :)) x ((++) xs) ys .

We conclude that

revcat = foldr (𝜆x f → f · (x :)) id .

13. Fold on natural numbers.

(a) The predicate even :: Nat → Bool yields True iff. the input is an even number. Define
even in terms of foldN .

Solution:

even = foldN not True .

(b) Express the identity function on natural numbers id n = n in terms of foldN .

Solution:

id = foldN 1+ 0 .

14. Fuse even into (+n). This way we get a function that checks whether a natural number is
even after adding n .

Solution: Recall that (+n) = foldN 1+ n . To fuse even · (+n) into one foldN , the base
value is even n . To find out the step function, recall that even (1+ n) = not (even n).
We may then conclude:

even · (+n) = foldN not (even n) .

15. The famous Fibonacci number is defined by:

fib 0 = 0
fib 1 = 1
fib (2 + n) = fib (1 + n) + fib n .

The definition above, when taken directly as an algorithm, is rather slow. Define fib2 n =

(fib (1+n), fib n). Derive an𝑂 (𝑛) implementation of fib2 by fusing it with id ::Nat → Nat.

Solution: Recall that id = foldN (1+) 0. Fusing fib2 into id , the base value is fib2 0 =

(1, 0). To construct the step function we calculate

fib2 (1+ n)
= (fib (1+ (1+ n)), fib (1+ n))

= { definition of fib }
(fib (1+ n) + fib n, fib (1+ n))

= (𝜆(x , y) → (x + y , x )) (fib2 n) .

Therefore we conclude that

fib2 = foldN (𝜆(x , y) → (x + y , x )) (1, 0) .

16. What are the fold fusion theorems for ETree and ITree?

Solution:

h · foldIT f e = foldIT g (h e) ⇐ h (f x y z ) = g x (h y) (h z ) ,

h · foldET f k = foldET g (h · k ) ⇐ h (f x y) = g (h x ) (h y) .

